

# Ordi-calc

## « Tablez » sur votre ZX

Visicalc est devenu, depuis son introduction, l'un des logiciels les plus populaires; mais sa complexité n'a permis jusqu'ici son adaptation que sur des micro-ordinateurs haut de gamme. C'est pourquoi nous vous présentons aujourd'hui Ordi-Calc, une version simplifiée de ce programme, adaptée aux micro-ordinateurs de faible capacité mémoire. En l'occurrence, un ZX 81 de Sinclair avec une extension mémoire de 16 Ko. Ce mois-ci, nous vous décrivons le « cœur » du programme Ordi-calc, c'est-à-dire sa version limitée à l'extrême. Il comprend les processus d'élaboration des tableaux, de chargement des valeurs et des relations, le programme de calcul, et les opérations de chargement et de sauvegarde. Cette version simplifiée vous permettra de vous familiariser avec Ordi-calc et de réaliser les calculs que vous souhaitez. Dans le prochain numéro vous seront proposées des extensions à ce logiciel, qui donneront toute sa puissance et sa souplesse d'utilisation à Ordi-calc.

Vous pourrez alors inclure des textes et des signes graphiques dans votre tableau, opérer des introductions répétitives (de valeurs de fonction ou de texte) paramétrées ou non, remettre des cases à blanc, ou utiliser Ordi-calc dans un programme Basic de votre création.

Ce logiciel est un utilitaire permettant de générer un tableau de valeurs, dont les éléments peuvent être liés les uns aux autres par un faisceau de relations, définies ici sous le nom de fonctions.

La modification de l'un des éléments du tableau provoque le recalcul automatique de tous les éléments qui leur sont liés par une fonction. L'ensemble se compose de deux tableaux Basic. Le premier, F\$(I,J) reçoit les valeurs ou les fonctions introduites dans Ordi-calc; le deuxième, T\$(I,J), reçoit les valeurs affichées à l'écran. Il est le résultat du calcul des fonctions figurant au tableau F\$(I,J) à l'aide d'une opération VAL F\$(I,J).

Ordi-calc permet donc, par exemple, d'établir des tableaux comptables et d'effectuer des simulations en modifiant la valeur de certains paramètres.

Il permet également de calculer les termes de progressions arithmétiques ou géométriques, de suites récurrentes, de sommer des séries ou de chercher les racines de certaines équations. Il peut, d'autre part, s'intégrer dans un programme Basic plus général, permettant de jouer avec la valeur des paramètres du tableau.

Le programme est lancé automatique-

```

500 PRINT AT 8,11;"*****"
510 PRINT AT 10,11;"ordi-calc"
520 PRINT AT 12,11;"*****"
530 PRINT AT 18,0;"P.ROUILLIER          SCIENCE ET VIE"
540 PAUSE 120
550 CLS
650 PRINT "VOULEZ VOUS GENERER UN TABLEAU ?"
660 INPUT R$
670 IF R$="0" THEN GOTO 1600
680 IF R$(">") THEN GOTO 660
690 LET Q=1
695 LET W$=""
700 LET B$=""
705 LET D$=" "
710 LET C$="--"
715 LET P$="-----"
720 LET BX=1
725 LET BY=1
730 LET CX=1
735 LET CY=1
740 LET W=1
745 LET Z=1
750 LET X=W
755 LET Y=Z
760 CLS
765 FAST
770 PRINT P$
775 PRINT
780 PRINT
785 PRINT.P$
790 FOR I=1 TO 17
795 PRINT D$
800 NEXT I
870 FOR J=Y TO Y+16/P
880 IF J>B THEN GOTO 942
890 FOR I=X TO X+D
900 IF I>A THEN GOTO 945
910 PRINT AT 0,(I-X)*(N+1)+2;STR$ I+C$(1 TO 2-LEN STR$ I)
915 PRINT AT (J-Y)*P+4,0;STR$ J+D$(1 TO 2-LEN STR$ J)
920 PRINT AT (J-Y)*P+4,(I-X)*(N+1)+2;T$(I,J);"*"
930 NEXT I
940 NEXT J
941 GOTO 950
942 PRINT AT (J-Y)*P+4,0;D$
944 GOTO 940
945 PRINT AT 0,(I-X)*(N+1)+2;C$
946 GOTO 930
950 GOSUB 1500
955 IF CX=BX AND BY=CY THEN GOTO 990
960 IF CX>X AND CX<=X+D AND CY>Y AND CY<=Y+16/P THEN
PRINT AT (CY-Y)*P+4,(CX-X)*(N+1)+2;T$(CX,CY)
970 LET CX=BX
980 LET CY=BY
990 SLOW
995 IF INKEY$="" THEN GOTO 995
1000 IF INKEY$="0" THEN LET Q=5
1010 IF INKEY$="P" THEN LET Q=1
1020 IF INKEY$="5" THEN LET BX=CX-Q
1030 IF INKEY$="8" THEN LET BX=CX+Q
1040 IF INKEY$="6" THEN LET BY=CY+Q
1050 IF INKEY$="7" THEN LET BY=CY-Q
1060 IF INKEY$="I" THEN GOTO 1250
1090 IF INKEY$="S" THEN GOTO 2400
1097 IF INKEY$="Z" THEN COPY
1100 IF BX<1 OR BX>A OR BY<1 OR BY>B THEN GOTO 1215
1120 FAST
1130 IF BX<X THEN LET W=X-Q
1140 IF BX>X+D THEN LET W=X+Q
1150 IF BY<Y THEN LET Z=Y-Q
1160 IF BY>Y+16/P THEN LET Z=Y+Q
1170 IF W=X AND Z=Y THEN GOTO 1200
1180 GOTO 750
1200 GOSUB 1500
1210 GOTO 955

```

## Commentaires

500-550 Ecran de présentation d'Ordicalc.

650-680 Création d'un nouveau tableau, ou rappel d'un tableau et de variables préalablement sauvegardées.

690-760 Initialisation des variables, le curseur est positionné dans la case en haut à gauche.

765-800 Impression du cadre général du tableau.

870-950 Impression proprement dite du tableau Ordicalc.

910 Numérotation des colonnes

915 Numérotation des lignes

920 Impression des valeurs et des séparateurs (\*)

955-990 Déplacement du curseur et calcul de ses nouvelles coordonnées.

995-1097 Commandes générales d'Ordicalc.

1100-1120 Détection d'une sortie de tableau.

1180-1210 Décalage de la fenêtre d'édition du tableau pour maintenir le curseur à l'écran.

1215-1220 Refus de déplacement du curseur pour empêcher toute sortie de tableau. Edition d'un message.

1250-1265 Introduction d'une valeur ou d'une fonction 1262 WARNING, indiquant que l'on cherche à introduire une fonction d'une longueur supérieure à la dimension décidée au départ.

1305-1410 Boucle de calcul du tableau. Les calculs sont effectués colonne après colonne, en partant de celle de plus bas indice (gauche) et en allant vers celle de plus fort indice (droite). Dans chaque colonne, les calculs sont effectués en partant de la ligne de plus bas indice (haut) et en allant vers celle de plus haut indice (bas).

1330 sélectionne les cases contenant des textes (voir numéro suivant) pour les éliminer dans la boucle de calcul.

1360 détecte les dépassements de capacité et bloque tous calculs sur les chiffres tronqués en inversant le dernier caractère.

1500-1570 Inversion du curseur. La visualisation du curseur est réalisée en imprimant la case où il se trouve en vidéo inversé (rectangle noir sur une case vide, nombre en vidéo inversée dans les autres cas).

1600-1890 Dimensionnement du tableau d'Ordicalc.

2400-2450 Sauvegarde du tableau d'Ordicalc et du jeu de variables.

```

1215 IF BX<1 THEN LET BX=CX
1216 IF BX>A THEN LET BX=CX
1217 IF BY<1 THEN LET BY=CY
1218 IF BY>B THEN LET BY=CY
1219 PRINT AT 1,0;"SORTIE DE TABLEAU"
1220 GOTO 990
1250 INPUT H$
1252 PRINT AT 1,25;"      "
1255 IF LEN H$>C THEN GOTO 1262
1257 LET F$(BX,BY)=H$
1260 GOTO 1305
1262 PRINT AT 1,25;"***F$***"
1265 GOTO 1250
1305 FAST
1310 FOR I=1 TO A
1320 FOR J=1 TO B
1330 IF F$(I,J,1)="$" THEN GOTO 1400
1340 IF F$(I,J)<1 TO 20)="$"      " THEN GOTO 1370
1350 LET G$=STR$ VAL F$(I,J)
1360 IF LEN G$>N THEN LET G$(N)=CHR$(CODE G$(N)+128)
1365 LET T$(I,J)=G$
1370 NEXT J
1380 NEXT I
1390 GOTO 750
1400 LET T$(I,J)=F$(I,J)(2 TO )
1410 GOTO 1370
1500 LET A$=T$(BX,BY)
1510 LET B$=""
1520 FOR I=1 TO LEN A$
1530 IF CODE A$(I)>128 THEN LET B$=B$+CHR$(CODE A$(I)+128)
1535 IF CODE A$(I)<128 THEN LET B$=B$+CHR$(CODE A$(I)-128)
1540 NEXT I
1550 PRINT AT (BY-Y)*P+4,(BX-X)*(N+1)+2;B$
1555 PRINT AT 1,0;W$
1556 PRINT AT 2,0;W$
1560 PRINT AT 1,0;"T$( ";BX;" ";BY;" )=";F$(BX,BY)(1 TO 40)
1570 RETURN
1600 CLS
1610 PRINT AT 0,6;"dimensions générales"
1620 LET A$="-----"
1625 PRINT A$
1630 PRINT "DIMENSIONS DU TABLEAU"
1640 PRINT "* COLONNES:";
1650 INPUT A
1660 PRINT A
1670 PRINT "* LIGNES...";
1680 INPUT B
1690 PRINT B
1695 PRINT
1700 PRINT "* FORMULES"
1705 PRINT "SIMPLES/40-COMPLEXES/80-:";
1710 INPUT C
1720 PRINT C
1730 PRINT
1740 PRINT "* TAILLES COLONNES:";
1750 INPUT N
1760 PRINT N
1770 PRINT A$
1780 PRINT "* OCCUPATION MEMOIRE:";A$B*(N+C)+5000
1785 PRINT
1790 LET D=INT(30/(N+1)-1)
1800 PRINT "* NB.COLONNES A L ECRAN:";D+1
1810 PRINT A$
1820 PRINT "* INTERLIGNE "
1830 PRINT "SIMPLE/1-DOUBLE/2-:";
1840 INPUT P
1850 PRINT P
1855 PRINT A$
1860 DIM T$(A,B,N)
1870 DIM F$(A,B,C)
1875 SLOW
1880 IF INKEY$="" THEN GOTO 1880
1890 GOTO 690
2400 CLS
2410 PRINT "POSITIONNEZ LA BANDE ET FAITES  NEW LINE"
2420 INPUT A$
2430 CLS
2440 SAVE "ORDI-CALC"
2450 GOTO 500

```

### Ne cachez plus vos talents...

Envoyez-nous un programme que vous avez écrit et peut-être recevrez-vous une bourse de 800 F. Chaque mois en première page du Cahier des Programmes, nous publierons le programme de l'un de nos lecteurs. Vous devez nous faire parvenir un listing complet du programme, une brève description de ses fonctionnalités, une photographie de vous et bien sûr une disquette ou une cassette. Envoyez le tout à SVM, 5, rue de la Baume, 75008 Paris.

En précisant vos nom, adresse et numéro de téléphone.

ment lors de son chargement. Au cas où un arrêt quelconque (BREAK) pourrait vous amener à relancer le programme, faites-le par GOTO 500 et surtout pas par RUN, qui effacerait toutes les variables de la mémoire centrale.

On doit définir au préalable le nombre de colonnes ou de lignes souhaitées ; la largeur des colonnes ; la longueur des chaînes de caractère servant à définir les fonctions. Attention : Seules seront affichées à l'écran, en haut du tableau, les 40 premières positions de la fonction. L'ordinateur vous indique alors la place occupée en mémoire par le programme et les tableaux. Vous pouvez ainsi évaluer l'espace disponible pour introduire votre programme Basic. Il vous indique également le nombre de colonnes visibles conjointement sur l'écran.

Vous pourrez ensuite choisir le mode d'impression de votre tableau, en double interligne (9 lignes par écran) ou en simple interligne (17 lignes par écran). L'ordinateur s'arrête alors, et vous pouvez obtenir l'impression de votre tableau en tapant sur n'importe quelle touche.

### Déplacement du curseur


Au premier affichage du tableau, le curseur est positionné dans la case (1,1) en haut à gauche du tableau.

Il se déplace :

- vers la droite par la touche 8
- vers la gauche par la touche 5
- vers le haut par la touche 7
- vers le bas par la touche 6

En appuyant sur la touche «Q», le curseur se déplace de cinq cases à la fois. En appuyant sur la touche «P», on restaure le mouvement case par case.

Tout déplacement qui entraînerait le curseur au-delà des dimensions du tableau est refusé, et le message «SORTIE DE TABLEAU» est affiché.



Un déplacement du curseur vers une case située en dehors de l'affichage de l'écran entraîne un déplacement du tableau pour rendre cette case visible.

### Introduction de variables ou de fonctions

Positionnez le curseur sur la case dans laquelle vous voulez réaliser l'introduction. Appuyez sur la touche «I» ; l'ordinateur attend alors l'introduction d'une chaîne de caractères.

\* Une valeur numérique est introduite


en la tapant directement sur le clavier.

\* Une fonction sera introduite sous la forme d'une fonction Basic, respectant la syntaxe de ce langage. Si un élément du tableau est utilisé en paramètre dans cette fonction, il sera désigné par VAL T\$, suivi du numéro de la colonne et de celui de la ligne, sous la forme générale VAL T\$(c,l).

La valeur de la fonction, calculée par VAL (fonction), sera introduite dans la case désignée par le curseur.

### Dépassement de capacité

Si une valeur à afficher à l'écran dépasse la taille d'une colonne, le résultat est tronqué à la taille de celui-ci et le dernier chiffre est affiché en vidéo inversée. Cette valeur ne peut en aucun cas être utilisée dans un calcul, mais elle peut rester affichée à l'écran sans inconvénient.



Si une fonction introduite dépasse le dimensionnement choisi (40 ou 80 caractères) le signal \*\* F\$ \*\* est affiché en haut à gauche, et l'introduction rejetée.

### Sauvegarde - chargement

Après avoir défini un tableau et en avoir rempli les diverses cases, il est souvent intéressant de sauvegarder l'ensemble sur cassette. Pour cela, appuyez sur la touche «S». L'ordinateur affiche alors : «POSITIONNEZ LA BANDE ET FAITES NEW LINE».

Mettez la cassette vierge en place et le magnétophone en enregistrement. Appuyez sur la touche «NEW LINE». Le programme et les valeurs figurant dans le tableau sont sauvegardés.

L'ordinateur revient ensuite automatiquement au début du programme.

Le chargement s'effectue normalement par la commande LOAD « » ou LOAD «ORDI-CALC». Après chargement, le programme se lance automatiquement.

Attention : Si vous avez chargé un tableau contenant des valeurs, répondez «N» (non) à la demande «VOULEZ-VOUS GENERER UN TABLEAU ?», faute de quoi vous risqueriez d'effacer toutes ces valeurs. Enfin, pour obtenir une copie du tableau figurant sur l'écran, appuyez sur la touche «Z».

Patrick ROUILLER

### COMMANDES GÉNÉRALES D'ORDI-CALC

#### Déplacement du curseur

- P LENT
- Q RAPIDE
- 8 A DROITE
- 5 A GAUCHE
- 6 EN HAUT
- 7 EN BAS

#### Fonctions

- I INTRODUCTION DE VALEURS OU DE FONCTIONS
- L INTRODUCTION DE TEXTES (\*)
- R REPÉTITIONS (\*)
- S SAUVEGARDE
- A SORTIE VERS UN PROGRAMME BASIC (\*)
- B REMISE A BLANC D'UNE CASE (\*)
- Z COPIE D'ECRAN SUR L'IMPRIMANTE

Les commandes suivies d'un (\*) ne sont pas disponibles avec la première partie d'Ordi-calc présentée ici ; elles seront détaillées dans notre prochain numéro.

#### LISTE DES VARIABLES

- A Nombre de colonnes dans le tableau
- B Nombre de lignes dans le tableau
- C Taille maximale d'une fonction
- D+1 Nombre de colonnes visibles à l'écran
- N Taille d'une colonne
- P Interligne (simple ou double)
- Q Pas de déplacement du curseur
- W Numéro de la colonne affichée en haut à gauche de l'écran
- X Variable pour l'affichage des colonnes à l'écran
- Y Variable pour l'affichage des lignes à l'écran
- Z Numéro de la ligne affichée en haut et à gauche de l'écran
- K1 Limite de répétitions
- K2 Limite de répétitions
- BX Numéro de la colonne centrale dans le déplacement du curseur
- BY Numéro de la ligne centrale dans le déplacement du curseur
- CX Numéro de la colonne finale dans le déplacement du curseur
- CY Numéro de la ligne finale dans le déplacement du curseur
- A\$ Chaîne intermédiaire pour l'inversion du curseur
- B\$ Chaîne intermédiaire pour l'inversion du curseur
- C\$ Chaîne graphique pour la présentation du tableau
- D\$ Chaîne graphique pour la présentation du tableau
- H\$ Chaîne intermédiaire pour l'introduction des valeurs, fonctions ou texte
- R\$ Réponses aux questions de l'ordinateur
- P\$ Chaîne graphique pour la présentation du tableau
- W\$ Chaîne de blancs pour effacer les messages
- F\$ (I,J) Tableau des valeurs ou des fonctions et textes introduits
- T\$ (I,J) Tableau des valeurs affichées à l'écran, et calculées à partir de F\$ (I,J)